Изчертаване на отсечки и окръжности

procedure DrawLine(x1,y1,x2,y2:integer);
   { Използва уравнението на права y=mx+b }
   var x,dx,dy:integer;

       m,b,y:real;

   begin

      dx:=x2-x1;

      dy:=y2-y1;

      m:=dy/dx;

      b:=y1-m*x1;

      x:=x1;

      repeat

        y:=m*x+b;

        PutPixel(x,round(y),Color);

        x:=x+1;

      until x>x2;

end;
procedure DDA(x1,y1,x2,y2:integer);

   { Цифров дифиренциален анализатор }
   var Diff,Error:real;

       x,y:integer;

   begin

     Diff:=(y2-y1)/(x2-x1);

     x:=x1; y:=y1;

     PutPixel(x,y,Color);

     Error:=0;

     while x<x2 do

       begin

         x:=x+1;

         Error:=Error+Diff;

         if Error>=0.5 then

           begin

             y:=y+1;

             Error:=Error-1;

           end;

         PutPixel(x,y,Color);

       end;

end;

procedure BresenhamI(x1,y1,x2,y2:integer);
   { Алгоритъм на Брезенам за изчертаване на отсечки в I октант }
    var Error,x,y:integer;

    begin

      x:=x1; y:=y1;

      PutPixel(x,y,Color);

      Error:=-(dx div 2);

      while x<x2 do

        begin

          x:=x+1;

          Error:=Error+dy;

          if Error>=0 then

              begin

                y:=y+1;

                Error:=Error-dx;

              end;

          PutPixel(x,y,Color);

        end;

end;

procedure DrawCircle(xc, yc, r : integer);

 { Апроксимация на окръжност с отсечки                                   }

 { xc и yc са координати на центъра на окръжността,                      }

 { а r е радиусът й                                                      }

  const thinc = 2.0*Pi / 100; { Апроксимация със 100 отсечки             }

  var   x1, y1,               { Координати на първата точка на отсечката }

        x2, y2 : integer;     { Координати на втората точка на отсечката }

        i              : integer;

        th             : real;

  begin

    th := 0;

    x1 := xc+r;

    y1 := yc;

    for i := 1 to 100 do

      begin

        th := th + thinc;

        x2 := xc + round(r*cos(th));

        y2 := yc + round(r*sin(th)* AspectRatio);

        Line(x1, y1, x2, y2);

        x1 := x2; y1 := y2;

      end;

  end;

procedure CircleMidPoint(xc, yc, r : integer);

   { Изчертаване на окръжности по метода на средната точка }
    var x, y, p : integer;

    procedure PlotPoints ( x, y : integer);

      begin

        PutPixel(xc + x, yc + y, Color);

        PutPixel(xc - x, yc + y, Color);

        PutPixel(xc + x, yc - y, Color);

        PutPixel(xc - x, yc - y, Color);

        PutPixel(xc + y, yc + x, Color);

        PutPixel(xc - y, yc + x, Color);

        PutPixel(xc + y, yc - x, Color);

        PutPixel(xc - y, yc - x, Color);

      end;

    begin

      x:=0;

      y:=r;

      p:=1-r;

      plotPoints(x,y);

      while x<y do

        begin

          x:= x+1;

          if p<0 then  p:= p + 2*x +1

                 else  begin

                         y := y-1;

                         p:= p + 2*(x-y) +1

                       end;

          plotPoints(x,y);

        end;

      end;

